

ÜBUNG [01]: Wiederholung: Lineare Algebra II

Wiederhole die Themen der linearen Algebra aus diesem Semester, wir stellen Fragen im Testat!

Affine Gruppe

[Aufgaben: 6

[> **restart**;

[> **with(LinearAlgebra)**;

Affiner Raum - Definitionen

Die affine Geometrie kann weitgehend als das Studium diverser Operationen der affinen Gruppe verstanden werden.

MATH: Sei V ein K -Vektorraum. Ein **affiner Raum** über V ist eine nicht leere Menge A , genannt Punktmenge, auf der V regulär operiert. Genauer ist ein affiner Raum ein Tripel (A, V, τ) , wobei $\tau: V \times A \rightarrow A$ eine reguläre Operation von V auf A ist. Hierbei heißt **V Translationsraum** von A (Bezeichnung: $T(A)$).

BEISPIEL:

Wir möchten im n -dimensionalen (die Definition der Dimension folgt unten) affinen Raum rechnen und bedienen uns dabei des folgenden Modells: Wir schreiben die Punkte des n -dimensionalen affinen Raumes über dem Körper K als Spaltenvektoren in $K^{(n+1) \times 1}$, deren letzter Eintrag gleich 1 ist.

Dies ist ein affiner Raum wie uns der folgende Satz sagt:

MATH: Sei V ein $(n+1)$ -dimensionaler K -Vektorraum und $0 \neq \varphi: V \rightarrow K$ ein lineares Funktional. Dann ist die Faser von 1, also

$$V_1 := \{p \in V \mid \varphi(p) = 1\}$$

das Modell eines n -dimensionalen affinen Raumes und

$$\text{Kern}(\varphi) = \{v \in V \mid \varphi(v) = 0\}$$

ist der Translationsraum von V_1 . Hierbei ist die Operation $\text{Kern}(\varphi)$ auf V_1 durch die Addition der Vektoren gegeben.

Dieses Modell bezeichnen wir mit $A_n(K)$.

BEZEICHNUNG: Die Elemente des affinen Raumes heißen **Punkte**, die des Translationsraumes **Vektoren**. Die Vektoren tun etwas mit den Punkten: Sie verschieben.

MATH: Sind A und B Punkte des affinen Raumes (A, V, τ) , so gibt es einen eindeutigen Vektor $v \in V$ mit: $\tau(A, v) = B$. (**DENKANSTOSS:** Warum?)

Wir bezeichnen dieses v mit \overrightarrow{AB} und nennen v den Translationsvektor von A nach

B.

Immer wenn wir neue Strukturen einführen sollten wir uns fragen, was die Homomorphismen dieser sind. Sprich welche Abbildungen erhalten unsere Struktur?

MATH: Es seien $A_1 = (P_1, T_1, \tau_1)$ und $A_2 = (P_2, T_2, \tau_2)$ affine Räume über dem Körper K . Eine Abbildung $f: P_1 \rightarrow P_2$ heißt **affine Abbildung**, falls eine *lineare* Abbildung $\bar{f}: T_1 \rightarrow T_2$ existiert mit $f(P)f(Q) = \bar{f}(\overrightarrow{PQ})$ für alle $P, Q \in P_1$. Die Abbildung \bar{f} heißt **linearer Anteil** von f .

Die affine Abbildung f ist injektiv, surjektiv resp. bijektiv genau dann, wenn \bar{f} injektiv, surjektiv resp. bijektiv ist. (**DENKANSTOSS:** Warum?)

Die Menge aller bijektiven affinen Selbstabbildungen eines affinen Raumes A heißt die **affine Gruppe** von A , kurz $Aff(A)$.

DENKANSTOSS: Verifiziere die Gruppenaxiome für $Aff(A)$.

MATH: Zwei affine Räume $A_1 = (P_1, T_1, \tau_1)$ und $A_2 = (P_2, T_2, \tau_2)$ über demselben Körper K sind genau dann **affin isomorph**, d.h. es existiert eine bijektive affine Abbildungen zwischen ihnen, wenn gilt:

$$\dim(T_1) = \dim(T_2).$$

Man nennt $\dim(A_1) := \dim(T_1)$ die **Dimension** des affinen Raumes A_1 .

Ein affiner Isomorphismus $A_1 \rightarrow A_{\dim(A_1)}(K)$ heißt **affines Koordinatensystem**.

Die Menge der Punkte eines affinen Raumes kann eine (fast) beliebige Menge sein, sie erhält ihre Struktur erst durch die Operation des Vektorraumes. Wir betrachten ein warnendes Beispiel:

```
> f := x -> x^19+3*x^18+2*x^17+2*x^13+6*x^12+4*x^11-2*x^3-5*x^2-x+3 mod 3;
```

```
f:=x->(x^19+3*x^18+2*x^17+2*x^13+6*x^12+4*x^11-2*x^3-5*x^2-x+3) mod 3 (1.1.1)
```

```
> g := x -> Determinant(Matrix(4,4,[0,1,x,2,2,0,0,1,1,2,0,1,0,1,0,1]))+1 mod 3;
```

```
g:=x->(LinearAlgebra:-Determinant(Matrix(4,4,[0,1,x,2,2,0,0,1,1,2,0,1,0,1,0,1]))+1) mod 3 (1.1.2)
```

```
> h := x -> x^21-7*x^19+21*x^17-35*x^15+35*x^13-21*x^11+7*x^9-x^7+x^2+1 mod 3;
```

```
h:=x->(x^21-7*x^19+21*x^17-35*x^15+35*x^13-21*x^11+7*x^9-x^7+x^2+1) mod 3 (1.1.3)
```

Die obigen Funktionen seien Abbildungen von $\mathbb{F}_3 \rightarrow \mathbb{F}_3$.

ÜBUNG [02]:

Hinweis: Philosophiere über eine gute Reihenfolge der Aufgaben.

Es sei $K := \mathbb{F}_3$. Es sei $V := \langle f, g \rangle \subseteq \mathbb{F}_3^3$ ein K -Vektorraum.

1.) Bestimme die Dimension d von V .

Ferner sei

$M := \{ \text{Gruppe1}, \text{Gruppe2}, \text{Gruppe3}, \text{Gruppe4}, \text{Gruppe5}, \text{Gruppe6}, \text{Gruppe7}, \text{Gruppe8}, \text{Gruppe9} \}$.

Es operiere V treu auf der Menge M via τ gegeben durch folgende Tabelle (Du musst nicht überprüfen, dass dies eine treue Operation ist):

g	Gruppe1	Gruppe2	Gruppe3	Gruppe4	Gruppe5	Gruppe6	Gruppe7	Gruppe8	Gruppe9
f	Gruppe1	Gruppe2	Gruppe3	Gruppe4	Gruppe5	Gruppe6	Gruppe7	Gruppe8	Gruppe9
M	Gruppe1	Gruppe2	Gruppe3	Gruppe4	Gruppe5	Gruppe6	Gruppe7	Gruppe8	Gruppe9

2.) Zeige, dass $\mathbb{A} := (M, V, \tau)$ ein affiner Raum über K ist.

3.) Bestimme den Translationsvektor von Gruppe9 zu Gruppe1 als Element von V .

4.) Es sei $\alpha(f) = g$ und $\alpha(g) = h$. Zeige, dass sich α zu einem wohldefinierten Endomorphismus von V fortsetzen lässt.

5.) Es sei durch $F(\text{Gruppe6}) = \text{Gruppe2}$ und den linearen Anteil α eine affine Abbildung gegeben. Bestimme das Bild von Gruppe7 unter F .

6.) Bestimme einen Isomorphismus von V nach $K^{d \times 1}$.

7.) Bestimme ein affines Koordinatensystem, also einen affinen Isomorphismus

von M nach $\mathbb{A}_d(\mathbb{F}_3)$, wobei deine Gruppe auf $\begin{bmatrix} 0 \\ \dots \\ 0 \\ 1 \end{bmatrix}$ abgebildet wird.

Hinweis: Philosophiere über eine gute Reihenfolge der Aufgaben.

MATH: Ein n -Tupel von Punkten (P_1, \dots, P_n) heißt **affin unabhängig** bzw. **abhängig**, wenn die Translationsvektoren $(\overrightarrow{P_1P_2}, \overrightarrow{P_1P_3}, \dots, \overrightarrow{P_1P_n})$ linear unabhängig bzw. abhängig sind.

Im Modell $A_n(K)$ ist dies einfacher:

Punkte aus V_1 sind **affin unabhängig** bzw. **abhängig**, wenn sie linear unabhängig bzw. abhängig in V sind.

MATH: Analog zum Erzeugnis im Vektorraumfall definiert man das **affine Erzeugnis** $\langle P \rangle_a$ für $P \in \mathbb{A}^n$ als den kleinsten affinen Raum, welcher die Punkte aus P enthält. Ein **affiner Teilraum** von $A := (\mathbb{A}, V, \tau)$ ist ein affiner Raum $B := (\mathbb{B}, W, \sigma)$ mit $\mathbb{B} \subseteq \mathbb{A}$, $W \leq V$ und $\sigma(w, p) = \tau(w, p)$ für alle $w \in W$, $p \in \mathbb{B}$.

DENKANSTOSS: Sei $A := (\mathbb{A}, V, \tau)$ ein affiner Raum. Ferner seien $U \leq V$ ein Teilraum des Translationsraumes und $p \in \mathbb{A}$ ein Punkt. Die Menge $\tau(U, p) := \{\tau(u, p) \mid u \in U\}$ ist ein affiner Teilraum von A .

MATH: Drei Punkte A, B, C heißen **kollinear**, falls sie affin abhängig sind, d.h. ein $\lambda \in \mathbb{R}$ existiert mit $\lambda \overrightarrow{AB} = \overrightarrow{AC}$. λ ist in diesem Fall eindeutig und heißt dann **Teilverhältnis** (kurz $TV(A, B, C)$ genannt) von (A, B, C) .

▼ Affine Gruppe

MATH: Die affine Gruppe $\text{Aff}(V_1)$ von V_1 ist der Stabilisator

$$\text{Stab}_{\text{GL}(V)}(\varphi) \leq \text{GL}(V)$$

von $\varphi \in V^*$ in $\text{GL}(V)$ (bei der Operation von $\text{GL}(V)$ auf V^*).

Wir kehren zu unserem Modell vom Anfang zurück: Wir betrachten $K^{4 \times 1}$ und das lineare Funktional φ ist die Projektion auf die letzte Komponente. Die Punkte V_1 sind die Elemente mit letzter Komponente 1 und die Vektoren des Translationsraumes $T(V_1) = \text{Kern}(\varphi)$ sind die mit letzter Komponente 0.

In diesem Modell bedeutet die Definition also folgendes: Wir suchen den Stabilisator der Projektion auf die letzte Komponente:

```
> v := <0|0|0|1>;
M := Matrix(4,symbol=a);
s := solve(convert(v.M^(-1)-v, list));
subs(s, M);
v := 'v':M := 'M':s := 's':
```

$$v := \begin{bmatrix} 0 & 0 & 0 & 1 \end{bmatrix}$$

$$M := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} & a_{1,4} \\ a_{2,1} & a_{2,2} & a_{2,3} & a_{2,4} \\ a_{3,1} & a_{3,2} & a_{3,3} & a_{3,4} \\ a_{4,1} & a_{4,2} & a_{4,3} & a_{4,4} \end{bmatrix}$$

$$s := \{a_{1,1} = a_{1,1}, a_{1,2} = a_{1,2}, a_{1,3} = a_{1,3}, a_{1,4} = a_{1,4}, a_{2,1} = a_{2,1}, a_{2,2} = a_{2,2}, a_{2,3} = a_{2,3}, a_{2,4} = a_{2,4}, a_{3,1} = a_{3,1}, a_{3,2} = a_{3,2}, a_{3,3} = a_{3,3}, a_{3,4} = a_{3,4}, a_{4,1} = 0, a_{4,2} = 0, a_{4,3} = 0, a_{4,4} = 1\}$$

$$\begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} & a_{1,4} \\ a_{2,1} & a_{2,2} & a_{2,3} & a_{2,4} \\ a_{3,1} & a_{3,2} & a_{3,3} & a_{3,4} \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

(1.2.1)

Also können wir die affine Gruppe in unserem Modell mittels dem folgendem Programm erstellen:

```
> Aff:=proc(n::posint,s)
  local M;
  M:=Matrix(n,n+1,symbol=s);
  M:=Matrix(<M,Matrix(1,n+1)>);
  M[n+1,n+1]:=1;
  return M;
end proc;
> S:=Aff(3,s);
```

$$S := \begin{bmatrix} s_{1,1} & s_{1,2} & s_{1,3} & s_{1,4} \\ s_{2,1} & s_{2,2} & s_{2,3} & s_{2,4} \\ s_{3,1} & s_{3,2} & s_{3,3} & s_{3,4} \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

(1.2.2)

Wir halten im Hinterkopf, ohne es Maple mitzuteilen, dass diese Matrix invertierbar sein soll.

```
> Determinant(S)<>0;
```

$$s_{1,1} s_{2,2} s_{3,3} - s_{1,1} s_{2,3} s_{3,2} - s_{1,2} s_{2,1} s_{3,3} + s_{1,2} s_{2,3} s_{3,1} + s_{1,3} s_{2,1} s_{3,2} - s_{1,3} s_{2,2} s_{3,1} \neq 0$$

(1.2.3)

DENKANSTOSS: $\text{Aff}(V_1)$ operiert auf V_1 durch

$$\text{Aff}(V_1) \times V_1 \rightarrow V_1: (g, P) \mapsto g(P)$$

d. h. das Anwenden der linearen Abbildungen aus $\text{Aff}(V_1)$ auf Elemente von V_1 liefert wieder Elemente von V_1 . Die induzierte Abbildung von $g \in \text{Aff}(V_1)$ auf V_1

ist einfach die Einschränkung von g auf V_1 .

Beachte: Die Operation ist **treu**, d. h. zwei verschiedene Elemente von $\text{Aff}(V_1)$ induzieren auch verschiedene Abbildungen von V_1 auf sich.

MATH: Die Begriffsbildungen der **affinen Geometrie** sollten also mit der Operation der affinen Gruppe verträglich sein:

- 1) affine Unabhängigkeit oder Abhängigkeit von Punkten sollte bei Anwendung von Elementen von $\text{Aff}(V_1)$ erhalten bleiben,
- 2) affine Teilräume sollten durch die Elemente der affinen Gruppe wieder in affine Teilräume abgebildet werden,
- 3) affine Abbildungen sollten durch Komposition mit Elementen von $\text{Aff}(V_1)$ wieder affine Abbildungen ergeben.

Wir sehen, dass $\text{Aff}(V_1)$ transitiv auf V_1 operiert:

> **P:=Vector(4,[0,0,0,1]);**

$$P:= \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \quad (1.2.4)$$

> **S.P;**

$$\begin{bmatrix} s_{1,4} \\ s_{2,4} \\ s_{3,4} \\ 1 \end{bmatrix} \quad (1.2.5)$$

Man sieht hier, dass man für jeden Wert für $s_{1,4}$, $s_{2,4}$, $s_{3,4}$ ein S findet, das invertierbar ist, daher folgt Transitivität.

Wir bestimmen den Stabilisator von P :

> **sSP:=[op(solve(convert(P-S.P,set)))];**

$$sSP:= [s_{1,4} = 0, s_{2,4} = 0, s_{3,4} = 0] \quad (1.2.6)$$

> **SP:=subs(sSP,S);**

$$SP:= \begin{bmatrix} s_{1,1} & s_{1,2} & s_{1,3} & 0 \\ s_{2,1} & s_{2,2} & s_{2,3} & 0 \\ s_{3,1} & s_{3,2} & s_{3,3} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.2.7)$$

Wir sehen, dass der Stabilisator als Gruppe isomorph zu $GL(3, K)$ ist und auch auf V_1 operiert wie $GL(3, K)$ auf $K^{3 \times 1}$ (sprich $\text{Kern}(\varphi)$). Also ist die affine

Geometrie eines affinen Raumes, in welchem ein Punkt ausgezeichnet ist, mit der Geometrie eines Vektorraumes gleichzusetzen. Z. B. ist der Stabilisator des Punktes P noch transitiv auf $V_1 - \{P\}$, so wie $GL(n, K)$ transitiv auf $K^{n \times 1} - \{0\}$ ist.

> **SP.P, SP.<1,0,0,1>;**

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} s_{1,1} \\ s_{2,1} \\ s_{3,1} \\ 1 \end{bmatrix} \quad (1.2.8)$$

Dies kann man auch so interpretieren: $\text{Aff}(V_1)$ hat auf $V_1 \times V_1$ genau zwei Bahnen: Die Diagonale

$$\Delta := \{(P, P) \mid P \in V_1\}$$

und den Rest

$$V_1 \times V_1 - \Delta.$$

Also sind die Paare zweier verschiedener Punkte **geometrisch nicht zu unterscheiden**. (Das wird sich in der Euklidischen affinen Geometrie ändern: Dort ist der Abstand eine Invariante der Gruppenoperation.)

Wir wollen noch den Stabilisator eines anderen Punktes bestimmen:

> **T:=Matrix(4,4)+1: T[1,4]:=1: T;**

$$\begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.2.9)$$

> **T.P;**

$$\begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \end{bmatrix} \quad (1.2.10)$$

> **T.SP.T^(-1);**

$$\begin{bmatrix} s_{1,1} & s_{1,2} & s_{1,3} & -s_{1,1} + 1 \\ s_{2,1} & s_{2,2} & s_{2,3} & -s_{2,1} \\ s_{3,1} & s_{3,2} & s_{3,3} & -s_{3,1} \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.2.11)$$

Dies sollte der Stabilisator von T.P sein:

> (T.SP.T^(-1)).(T.P);

$$\begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \end{bmatrix}$$

(1.2.12)

Wir wollen den Durchschnitt der beiden Stabilisatoren bestimmen. Um die Gleichungen hinschreiben zu können, müssen wir bei einer der Matrizen die Namen der Variablen verändern:

> Diffe:=subs(s=t,SP)-T.SP.T^(-1);

$$Diffe := \begin{bmatrix} t_{1,1} - s_{1,1} & t_{1,2} - s_{1,2} & t_{1,3} - s_{1,3} & s_{1,1} - 1 \\ t_{2,1} - s_{2,1} & t_{2,2} - s_{2,2} & t_{2,3} - s_{2,3} & s_{2,1} \\ t_{3,1} - s_{3,1} & t_{3,2} - s_{3,2} & t_{3,3} - s_{3,3} & s_{3,1} \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

(1.2.13)

> Gle:=map(i->op(map(j->Diffe[i,j],[\\$1..4])),{ \\$1..3});

Gle:= {s_{2,1}, s_{3,1}, s_{1,1} - 1, t_{1,1} - s_{1,1}, t_{1,2} - s_{1,2}, t_{1,3} - s_{1,3}, t_{2,1} - s_{2,1}, t_{2,2} - s_{2,2}, t_{2,3} - s_{2,3}, t_{3,1} - s_{3,1}, t_{3,2} - s_{3,2}, t_{3,3} - s_{3,3}}

(1.2.14)

> su:=[op(solve(Gle))];

su:= [s_{1,1} = 1, s_{1,2} = t_{1,2}, s_{1,3} = t_{1,3}, s_{2,1} = 0, s_{2,2} = t_{2,2}, s_{2,3} = t_{2,3}, s_{3,1} = 0, s_{3,2} = t_{3,2}, s_{3,3} = t_{3,3}, t_{1,1} = 1, t_{1,2} = t_{1,2}, t_{1,3} = t_{1,3}, t_{2,1} = 0, t_{2,2} = t_{2,2}, t_{2,3} = t_{2,3}, t_{3,1} = 0, t_{3,2} = t_{3,2}, t_{3,3} = t_{3,3}]

(1.2.15)

> S2:=subs(su,T.SP.T^(-1));
Determinant(S2)<>0;

$$S2 := \begin{bmatrix} 1 & t_{1,2} & t_{1,3} & 0 \\ 0 & t_{2,2} & t_{2,3} & 0 \\ 0 & t_{3,2} & t_{3,3} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$t_{2,2} t_{3,3} - t_{2,3} t_{3,2} \neq 0$$

(1.2.16)

> S2.P,S2.(T.P);

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \end{bmatrix}$$

(1.2.17)

- 1.) Bestimme die Bahnen des Zweipunktstabilisators S_2 auf V_1 .
- 2.) Was hat dieses Ergebnis mit dem Teilverhältnis dreier kollinearere Punkte zu tun, von denen die ersten beiden verschieden sind? (Hinweis: Für gegebene Punkte $A \neq B \in V_1$ gibt es eine affine Abbildung g mit $g(A) = (0, 0, 0, 1)$ und $g(B) = (1, 0, 0, 1)$. Wie sieht diese Abbildung aus?)
- 3.) Folgere aus 1.): Was sind die Bahnen von $\text{Aff}(V_1)$ auf der Menge der Punkttripel, bei denen die ersten beiden Punkte affin unabhängig sind?

MATH: $\text{Aff}(V_1)$ operiert linear auf $\text{Kern}(\varphi) = T(V_1)$. Diese Operation ist nicht treu, denn alle Translationen, d.h. die affinen Abbildungen mit der Identität als linearem Anteil, operieren trivial.

> **S;**

$$\begin{bmatrix} s_{1,1} & s_{1,2} & s_{1,3} & s_{1,4} \\ s_{2,1} & s_{2,2} & s_{2,3} & s_{2,4} \\ s_{3,1} & s_{3,2} & s_{3,3} & s_{3,4} \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.2.18)$$

> **t:='t':**

> **Tr:=Vector(4,i->if i=4 then 0 else t[i] end if);**

$$Tr := \begin{bmatrix} t_1 \\ t_2 \\ t_3 \\ 0 \end{bmatrix} \quad (1.2.19)$$

> **S.Tr;**

$$\begin{bmatrix} s_{1,1} t_1 + s_{1,2} t_2 + s_{1,3} t_3 \\ s_{2,1} t_1 + s_{2,2} t_2 + s_{2,3} t_3 \\ s_{3,1} t_1 + s_{3,2} t_2 + s_{3,3} t_3 \\ 0 \end{bmatrix} \quad (1.2.20)$$

Hier ist eine Translation:

> **TR:=Matrix(4,4)+1:for i from 1 to 3 do TR[i,4]:=r[i] end do:**
TR;

(1.2.21)

$$\begin{bmatrix} 1 & 0 & 0 & r_1 \\ 0 & 1 & 0 & r_2 \\ 0 & 0 & 1 & r_3 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.2.21)$$

> TR.Tr;

$$\begin{bmatrix} t_1 \\ t_2 \\ t_3 \\ 0 \end{bmatrix} \quad (1.2.22)$$

Also ist die Operation der Translationsuntergruppe von $\text{Aff}(V_1)$ auf $\text{Kern}(\varphi)$ in der Tat trivial.

Die Translationen bilden eine Untergruppe.

ÜBUNG [04]:

Zeige: Die Translationsuntergruppe von $\text{Aff}(V_1)$ ist (als Gruppe) isomorph zu $\text{Kern}(\varphi)$.

Hinweis: Beachte: Die Translationen sind 4×4 Matrizen und die Elemente von $\text{Kern}(\varphi)$ sind 4×1 Vektoren!

Die letzte Übung motiviert eine neue Definition des affinen Teilraums:

Eine Bahn einer Untergruppe der Translationsuntergruppe auf V_1 heißt **affiner Teilraum** von V_1 .

Zum Abschluss wollen wir noch ein wenig rechnen:

ÜBUNG [05]:

1.) Bestimme eine invertierbare affine Abbildung $f: K^{4 \times 1} \rightarrow K^{4 \times 1}$ mit

$$f \left(\begin{bmatrix} 3 \\ 5 \\ 1 \\ 1 \end{bmatrix} \right) = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \quad \text{und} \quad f \left(\begin{bmatrix} 6 \\ 8 \\ 7 \\ 1 \end{bmatrix} \right) = \begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \end{bmatrix}.$$

2.) Bestimme $f\left(\begin{bmatrix} 5 \\ 7 \\ 5 \\ 1 \end{bmatrix}\right)$ und $f\left(\begin{bmatrix} 6 \\ 8 \\ 6 \\ 1 \end{bmatrix}\right)$. Welcher Unterschied fällt auf? Erkläre das Phänomen.

▼ Warum wackelt der Tisch?

In der Abstellkammer liegt ein affiner Tisch mit vier Füßen. Seine Füße sind bei

den Koordinaten $\begin{bmatrix} 12 \\ 4 \\ 3 \end{bmatrix}$, $\begin{bmatrix} 10 \\ 7 \\ 4 \end{bmatrix}$, $\begin{bmatrix} 13 \\ 8 \\ 0 \end{bmatrix}$ und $\begin{bmatrix} 11 \\ -1 \\ 6 \end{bmatrix}$ im \mathbb{R}^3 .

> $(T1, T2, T3, T4) := \langle 12, 4, 3, 1 \rangle, \langle 10, 7, 4, 1 \rangle, \langle 13, 8, 0, 1 \rangle, \langle 11, -1, 6, 1 \rangle;$

$$T1, T2, T3, T4 := \begin{bmatrix} 12 \\ 4 \\ 3 \\ 1 \end{bmatrix}, \begin{bmatrix} 10 \\ 7 \\ 4 \\ 1 \end{bmatrix}, \begin{bmatrix} 13 \\ 8 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 11 \\ -1 \\ 6 \\ 1 \end{bmatrix} \quad (1.3.1)$$

Wir stellen den Tisch auf den Boden $B = \left\{ \begin{bmatrix} a \\ b \\ 0 \end{bmatrix}, a, b \in \mathbb{R} \right\}$, indem wir zuerst den

Fuß $T1$ auf die Koordinate $\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$ stellen. Dies tun wir mit Hilfe der folgenden

affinen Abbildung:

> $Tr1 := \langle \langle 1, 0, 0, 0 \rangle | \langle 0, 1, 0, 0 \rangle | \langle 0, 0, 1, 0 \rangle | \langle -12, -4, -3, 1 \rangle \rangle;$

$$Tr1 := \begin{bmatrix} 1 & 0 & 0 & -12 \\ 0 & 1 & 0 & -4 \\ 0 & 0 & 1 & -3 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.3.2)$$

> $\text{map}(t \rightarrow Tr1.t, [T1, T2, T3, T4]);$

$$\left[\begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -2 \\ 3 \\ 1 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ 4 \\ -3 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ -5 \\ 3 \\ 1 \end{bmatrix} \right] \quad (1.3.3)$$

Nun bewegen wir $T2$ nach $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$, aber wollen dabei $T1$ fest lassen. Hierzu

bestimmen wir den Stabilisator von $T1$ in der affinen Gruppe:

> A := <Matrix(3,4,symbol=a),<0|0|0|1>>;

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} & a_{1,4} \\ a_{2,1} & a_{2,2} & a_{2,3} & a_{2,4} \\ a_{3,1} & a_{3,2} & a_{3,3} & a_{3,4} \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.3.4)$$

> A.<0,0,0,1>-<0,0,0,1>;

$$\begin{bmatrix} a_{1,4} \\ a_{2,4} \\ a_{3,4} \\ 0 \end{bmatrix} \quad (1.3.5)$$

Dies muss 0 sein, also:

**> S1 := subs([a[1,4]=0,a[2,4]=0,a[3,4]=0], A);
Determinant(S1);**

$$S1 := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} & 0 \\ a_{2,1} & a_{2,2} & a_{2,3} & 0 \\ a_{3,1} & a_{3,2} & a_{3,3} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$a_{1,1} a_{2,2} a_{3,3} - a_{1,1} a_{2,3} a_{3,2} - a_{1,2} a_{2,1} a_{3,3} + a_{1,2} a_{2,3} a_{3,1} + a_{1,3} a_{2,1} a_{3,2} - a_{1,3} a_{2,2} a_{3,1} \quad (1.3.6)$$

Diese Determinante muss natürlich immer $\neq 0$ sein, ansonsten zerstören wir noch den Tisch.

> S1.<0,0,0,1>;

(1.3.7)

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \quad (1.3.7)$$

Um den Tisch aufzustellen, müssen wir nun folgendes Gleichungssystem lösen:

> S1.(Tr1.T2)=<1,0,0,1>;

$$\begin{bmatrix} -2 a_{1,1} + 3 a_{1,2} + a_{1,3} \\ -2 a_{2,1} + 3 a_{2,2} + a_{2,3} \\ -2 a_{3,1} + 3 a_{3,2} + a_{3,3} \\ 1 \end{bmatrix} = \begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \end{bmatrix} \quad (1.3.8)$$

> convert(S1.(Tr1.T2)-<1,0,0,1>,list);

solve(%);

Tr2_ := subs(% , S1);

$[-2 a_{1,1} + 3 a_{1,2} + a_{1,3} - 1, -2 a_{2,1} + 3 a_{2,2} + a_{2,3}, -2 a_{3,1} + 3 a_{3,2} + a_{3,3}, 0]$

$\{a_{1,1} = a_{1,1}, a_{1,2} = a_{1,2}, a_{1,3} = 2 a_{1,1} - 3 a_{1,2} + 1, a_{2,1} = a_{2,1}, a_{2,2} = a_{2,2}, a_{2,3} = 2 a_{2,1} - 3 a_{2,2}, a_{3,1} = a_{3,1}, a_{3,2} = a_{3,2}, a_{3,3} = 2 a_{3,1} - 3 a_{3,2}\}$

$$Tr2_ := \begin{bmatrix} a_{1,1} & a_{1,2} & 2 a_{1,1} - 3 a_{1,2} + 1 & 0 \\ a_{2,1} & a_{2,2} & 2 a_{2,1} - 3 a_{2,2} & 0 \\ a_{3,1} & a_{3,2} & 2 a_{3,1} - 3 a_{3,2} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.3.9)$$

Wir möchten eine möglichst einfache Transformation, also setzen wir geschickte Werte ein:

> Tr2 := subs([a[2,2]=1, a[3,1]=-1, a[3,2]=0, a[2,1]=0, a[1,1]=0, a[1,2]=0], Tr2_);

Determinant(Tr2);

$$Tr2 := \begin{bmatrix} 0 & 0 & 1 & 0 \\ 0 & 1 & -3 & 0 \\ -1 & 0 & -2 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.3.10)$$

Wo sind unsere Tischbeine nun?

> map(t->Tr2.Tr1.t, [T1,T2,T3,T4]);

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -3 \\ 13 \\ 5 \\ 1 \end{bmatrix}, \begin{bmatrix} 3 \\ -14 \\ -5 \\ 1 \end{bmatrix}$$

(1.3.11)

ÜBUNG [06]:

1.) Bestimme den Stabilisator von $\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$ und $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$ in der affinen Gruppe.

2.) Stelle das dritte Tischbein auf $\begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$.

3.) Was ist der Stabilisator von $\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$, $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$ und $\begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$ in der affinen Gruppe? Was

tut er mit den Tischbeinen?

4.) Warum wackelt der Tisch?

Dieses Vorgehen (Wahl eines geeigneten affinen Isomorphismus) ist ein oft genutztes Verfahren für Beweise in der affinen Geometrie.

MATH: Es sei \mathbb{A} ein affiner Raum und $P \in \mathbb{A}^n$ affin unabhängig. Dann ist die

affine Abbildung definiert durch $\langle P \rangle_a \rightarrow K^{(n+1) \times 1} : P_i \mapsto \begin{bmatrix} e_i \\ 1 \end{bmatrix}$ ein affiner

Isomorphismus.

Wir wollen dieses Vorgehen in einer letzten Aufgabe nutzen, doch vorher brauchen wir noch eine Definition.

MATH: Ein affiner Teilraum der Dimension 1 heißt **Gerade**. Zwei Geraden heißen **parallel**, wenn sie denselben Translationsraum haben.

ÜBUNG [07]:

Beweise den Strahlensatz:

Es sei \mathbb{A} ein affiner Raum über \mathbb{Q} . Die Punkte $(P_0, P_1, P_2) \in \mathbb{A}^3$ seien affin unabhängig und ferner seien Punkte $Q_1 \in \langle P_0, P_1 \rangle_a$ sowie $Q_2 \in \langle P_0, P_2 \rangle_a$ mit $Q_1, Q_2 \neq P_0$ gegeben, sodass die Gerade $\langle P_1, P_2 \rangle_a$ parallel zu $\langle Q_1, Q_2 \rangle_a$ ist. Dann

gilt:

$$TV(P_0, P_1, Q_1) = TV(P_0, P_2, Q_2).$$